Chapter 33
The GLMPOWER Procedure (Experimental)

Chapter Contents

Chapter 33

The GLMPOWER Procedure (Experimental)

Overview

Power and sample size analysis optimizes the resource usage and design of a study, improving chances of conclusive results with maximum efficiency. The GLMPOWER procedure performs prospective analyses for linear models, with a variety of goals:

- determining the sample size required to get a significant result with adequate probability (power)
- characterizing the power of a study to detect a meaningful effect
- conducting what-if analyses to assess sensitivity of the power or required sample size to other factors

Here *prospective* indicates that the analysis pertains to planning for a future study. This is in contrast to *retrospective* analysis for a past study, which is not supported by the procedure.

The statistical analyses that are covered include contrasts and tests of fixed class effects in univariate linear models, optionally with covariates. The covariates can be continuous or categorical. Tests and contrasts involving random effects are not supported. For a variety of other types of statistical analyses, see Chapter 56, “The POWER Procedure.”

Input for PROC GLMPOWER includes the components considered in study planning:

- design (including subject profiles and their allocation weights)
- statistical model
- contrasts of class effects
- significance level (alpha)
- surmised response means for subject profiles (often called “cell means”)
- surmised variability
- power
- sample size

In order to identify power or sample size as the result parameter, you designate it by a missing value in the input. The procedure calculates this result value over one or more scenarios of input values for all other components.

You specify the design and the cell means using an *exemplary data set*, an data set of artificial values constructed to represent the intended sampling design and the surmised response means in the underlying population. You specify the model and contrasts using MODEL and CONTRAST statements similar to those in the GLM, ANOVA, REG, and MIXED procedures. You specify the remaining parameters with the POWER statement, which is similar to analysis statements in the POWER procedure.

In addition to tabular results, PROC GLMPOWER produces graphs. You can produce the most common types of plots easily with default settings and use a variety of options for more customized graphics. For example, you can control the choice of axis variables, axis ranges, number of plotted points, mapping of graphical features (such as color, line style, symbol, and panel) to analysis parameters, and legend appearance.

The GLMPOWER procedure is one of several tools available in SAS/STAT software for power and sample size analysis. PROC POWER covers a variety of more basic analyses such as *t* tests, equivalence tests, confidence intervals, binomial proportions, multiple regression, and one-way ANOVA. The Power and Sample Size application provides a user interface and implements many of the analyses supported in the procedures.

The remaining sections of this chapter describe how to use PROC GLMPOWER and discuss the underlying statistical methodology. The "Getting Started" section on page 1790 introduces PROC GLMPOWER with examples of power computation for a two-way analysis of variance. The "Syntax" section on page 1795 describes the syntax of the procedure. The "Details" section on page 1805 summarizes the methods employed by PROC GLMPOWER and provides details on several special topics. The "Examples" section on page 1811 illustrates the use of the GLMPOWER procedure with several applications.

For more discussion and examples on power and sample size analysis for linear models, refer to Castelloe and O'Brien (2001), O'Brien and Shieh (1992), Muller et al (1992), and O'Brien and Muller (1993). For additional discussion on general power and sample size concepts, refer to Castelloe (2000), Muller and Benignus (1992), and Lenth (2001).

Getting Started

Simple Two-Way ANOVA

This example demonstrates how to use PROC GLMPOWER to compute and plot power for each effect test in a two-way analysis of variance (ANOVA).

Suppose you are planning an experiment to study the effect of light exposure on the growth of two varieties of flowers. The planned data analysis is a two-way ANOVA with flower height (measured at two weeks) as the response and a full model including light exposure, flower variety, and their interaction. You want to calculate the power of each effect test using a balanced design with a total of 60 specimens (10 for each combination of exposure and variety) and $\alpha = 0.05$ for each test.

As a first step, create an *exemplary data set* describing your conjectures about the underlying population means. You believe that the mean flower height for each combination of variety and exposure level (i.e., for each design profile, or for each *cell* in the design) roughly follows Table 33.1.

Table 33.1. Mean Flower Height by Variety and Exposure

	Exposure		
Variety	1	2	3
1	14	16	21
2	10	15	16

The following statements create a data set Exemplary containing these cell means.

```
data Exemplary;
   input Variety $ Exposure $ Height;
   datalines;
         1   1  14
         1   2  16
         1   3  21
         2   1  10
         2   2  15
         2   3  16
   ;
run;
```

You also conjecture that the error standard deviation is about 5.

Use the DATA= option in the PROC GLMPOWER statement to specify Exemplary as the exemplary data set. Identify the class variables (Variety and Exposure) using the CLASS statement. Specify the model using the MODEL statement. Use the POWER statement to specify power as the result parameter and provide values for the other analysis parameters, error standard deviation and total sample size.

```
proc glmpower data=Exemplary;
   class Variety Exposure;
   model Height = Variety | Exposure;
   power
      stddev = 5
      ntotal = 60
      power  = .;
run;
```

The MODEL statement defines the full model including both main effects and the interaction. The POWER= option in the POWER statement identifies power as the result parameter with a missing value (.). The STDDEV= option specifies an error standard deviation of 5, and the NTOTAL= option specifies a total sample size of 60. The default value for the ALPHA= option sets the significance level to $\alpha = 0.05$.

Figure 33.1 shows the output.

```
                         The GLMPOWER Procedure
   Tests of contrasts and effects in the fixed-effects general linear model

                        Fixed Scenario Elements

                Dependent Variable                  Height
                Error Standard Deviation                 5
                Total Sample Size                       60
                Alpha                                 0.05
                Error Degrees of Freedom                54


                             Computed Power

                                           Test
               Index           Source        DF        Power

                   1    Variety               1     0.717653
                   2    Exposure              2     0.956741
                   3    Variety*Exposure      2     0.191447
```

Figure 33.1. Sample Size Analysis for Two-Way ANOVA

The power is about 0.72 for the test of the Variety effect. In other words, there is a probability of 0.72 that the test of the Variety effect will produce a significant result (given the assumptions for the means and error standard deviation). The power is 0.96 for the test of the Exposure effect and 0.19 for the interaction test.

Now, suppose you want to account for some of your uncertainty in conjecturing the true error standard deviation by evaluating the power at reasonable low and high values, 4 and 6.5. Specify both of these values with the STDDEV= option in the POWER statement. You also want to plot power against total sample sizes between 30 and 90. Use the PLOT statement with X=N to request a plot with sample size on the x-axis. (The result parameter, here power, is always plotted on the other axis). Use the MIN= and MAX= options in the PLOT statement to specify the sample size range.

```
proc glmpower data=Exemplary;
   class Variety Exposure;
   model Height = Variety | Exposure;
   power
      stddev = 4 6.5
      ntotal = 60
      power  = .;
   plot x=n min=30 max=90;
run;
```

Figure 33.2 shows the output, and Figure 33.3 shows the plot.

```
                         The GLMPOWER Procedure
     Tests of contrasts and effects in the fixed-effects general linear model

                          Fixed Scenario Elements

                Dependent Variable                Height
                Total Sample Size                     60
                Alpha                               0.05
                Error Degrees of Freedom              54


                              Computed Power

                                         Std     Test
           Index          Source         Dev       DF         Power

               1    Variety              4.0        1      0.886787
               2    Variety              6.5        1      0.496378
               3    Exposure             4.0        2      0.996471
               4    Exposure             6.5        2      0.792885
               5    Variety*Exposure     4.0        2      0.279508
               6    Variety*Exposure     6.5        2      0.130095
```

Figure 33.2. Sample Size Analysis for Two-Way ANOVA with Input Ranges



Figure 33.3. Plot of Power versus Sample Size for Two-Way ANOVA with Input Ranges

Figure 33.2 reveals that the power ranges from about 0.130 to 0.996 for the different effect tests and scenarios for standard deviation, with a sample size of 60. In Figure 33.3, the line style identifies the effect test, and the plotting symbol identifies the standard deviation. The locations of the plotting symbols identify actual computed powers; the curves are linear interpolations of these points. Note that the computed points in the plot occur at sample size multiples of 6, because there are 6 cells in the design (and by default, sample sizes are rounded to produce integer cell sizes).

Incorporating Contrasts, Unbalanced Designs, and Multiple Means Scenarios

Suppose you want to compute power for the two-way ANOVA described in "Simple Two-Way ANOVA," but you want to:

- Try an unbalanced sample size allocation with respect to Exposure, using twice as many samples for levels 2 and 3 as for level 1.
- Consider an additional, less optimistic scenario for the cell means, shown in Table 33.2.
- Test a contrast of Exposure comparing levels 1 and 3.

Table 33.2. Additional Cell Means Scenario

	Exposure		
Variety	1	2	3
1	15	16	20
2	11	14	15

To specify the unbalanced design and the additional cell means scenario, you can add two new variables to the exemplary data set (W for the sample size weights, and AddHeight for the new cell means scenario):

```
data Exemplary;
   input Variety $ Exposure $  Height  AddHeight  W;
   datalines;
         1   1   14  15  1
         1   2   16  16  2
         1   3   21  20  2
         2   1   10  11  1
         2   2   15  14  2
         2   3   16  15  2
   ;
run;
```

In PROC GLMPOWER, specify the name of the weight variable using the WEIGHT statement, and specify the name of the new means variable AddHeight as an additional dependent variable in the MODEL statement. Use the CONTRAST statement to specify the contrast as you would in PROC GLM. The following statements perform the sample size analysis.

```
proc glmpower data=Exemplary;
   class Variety Exposure;
   model Height AddHeight = Variety | Exposure;
   weight W;
   contrast 'Exposure=1 vs Exposure=3' Exposure 1 0 -1;
   power
      stddev = 5
```

```
        ntotal = 60
        power  = .;
run;
```

Figure 33.4 shows the output.

```
                        The GLMPOWER Procedure
   Tests of contrasts and effects in the fixed-effects general linear model

                       Fixed Scenario Elements

                Weight Variable                        W
                Error Standard Deviation               5
                Total Sample Size                     60
                Alpha                               0.05
                Error Degrees of Freedom              54


                           Computed Power

                                                        Test
       Index          Source              Dependent       DF        Power

           1     Exposure=1 vs Exposure=3    Height          1     0.950615
           2     Exposure=1 vs Exposure=3    AddHeight       1     0.705451
           3     Variety                     Height          1     0.672128
           4     Variety                     AddHeight       1     0.753735
           5     Exposure                    Height          2     0.911488
           6     Exposure                    AddHeight       2     0.633114
           7     Variety*Exposure            Height          2     0.216733
           8     Variety*Exposure            AddHeight       2     0.136705
```

Figure 33.4. Sample Size Analysis for More Complex Two-Way ANOVA

The power of the contrast of Exposure levels 1 and 3 is about 0.95 for the original cell means scenario (Height) and 0.71 for the new one (AddHeight). The power is higher for the test of Variety, but lower for the tests of Exposure and of Variety*Exposure for the new cell means scenario compared to the original one. Note also for the Height scenario that the power for the unbalanced design (Figure 33.4) compared to the balanced design (Figure 33.1) is slightly lower for the tests of Variety and Exposure, but slightly higher for the test of Variety*Exposure.

Syntax

The following statements are available in PROC GLMPOWER.

PROC GLMPOWER < *options* > **;**

CLASS *variables* **;**
MODEL *dependent-variables* = *effects* **;**

WEIGHT *variable* **;**

CONTRAST *'label' effect values* < . . . *effect values* > **;**

POWER < *options* > **;**

PLOT < *plot-options* > < **/** *graph-options* > **;**

The PROC GLMPOWER statement, the MODEL statement, and the POWER statement are required. If your model contains classification effects, the classification variables must be listed in a CLASS statement, and the CLASS statement must appear before the MODEL statement. In addition, CONTRAST statements must appear after the MODEL statement. PLOT statements must appear after the POWER statement defining the analysis for the plot.

You can use multiple CONTRAST, POWER, and PLOT statements. Each CONTRAST statement defines a separate contrast. Each POWER statement produces a separate sample size and uses the information contained in the CLASS, MODEL, WEIGHT, and all CONTRAST statements. Each PLOT statement refers to the previous POWER statement and generates a separate graph (or set of graphs).

Table 33.3 summarizes the basic functions of each statement in PROC GLMPOWER. The syntax of each statement in Table 33.3 is described in the following pages.

Table 33.3. Statements in the GLMPOWER Procedure

Statement	Description
PROC GLMPOWER	invokes procedure and specifies exemplary data set
CLASS	declares classification variables
CONTRAST	defines linear tests of model parameters
MODEL	defines model and specifies dependent variable(s) used for cell means scenarios
WEIGHT	specifies variable for allocating sample sizes to different subject profiles
POWER	identifies parameter to solve for and provides one or more scenarios for values of other analysis parameters
PLOT	constructs graphs for preceding POWER statement

PROC GLMPOWER Statement

PROC GLMPOWER < *options* > **;**

The PROC GLMPOWER statement invokes the GLMPOWER procedure. You can specify the following options.

DATA=*SAS-data-set*

names a SAS data set to be used as the exemplary data set, which is an artificial data set constructed to represent the intended sampling design and the conjectured response means for the underlying population.

PLOTONLY

specifies that only graphical results from the PLOT statement should be produced.

CLASS Statement

CLASS *variables* **;**

The CLASS statement names the classification variables to be used in the analysis. Classification variables can be either character or numeric.

CONTRAST Statement

CONTRAST *'label' effect values* < . . . *effect values* > **;**

The CONTRAST statement enables you to define custom hypothesis tests by specifying an $\mathbf{L}$ vector or matrix for testing the hypothesis $\mathbf{L}\boldsymbol{\beta} = 0$. Thus, to use this feature you must be familiar with the details of the model parameterization used in PROC GLM. For more information, see the "Parameterization of PROC GLM Models" section on page 1651 of Chapter 31, "The GLM Procedure." All of the elements of the $\mathbf{L}$ vector may be given, or if only certain portions of the $\mathbf{L}$ vector are given, the remaining elements are constructed by PROC GLM from the context (in a manner similar to rule 4 discussed in the "Construction of Least-Squares Means" section on page 1684 of Chapter 31, "The GLM Procedure,").

There is no limit to the number of CONTRAST statements you can specify. Each sample size analysis will include tests for all CONTRAST statements.

In the CONTRAST statement,

label identifies the contrast on the output. A label is required for every contrast specified. Labels must be enclosed in quotes.

effect identifies an effect that appears in the MODEL statement, or the INTERCEPT effect. You do not need to include all effects that are in the MODEL statement.

values are constants that are elements of the $\mathbf{L}$ vector associated with the effect.

The CONTRAST statement enables you to perform custom hypothesis tests. If the hypothesis is estimable, then the sum of squares due to it, SS($H_0\colon \mathbf{L}\boldsymbol{\beta} = 0$), is computed as

$$(\mathbf{Lb})'(\mathbf{L}(\mathbf{X}'\mathbf{X})^{-}\mathbf{L}')^{-1}(\mathbf{Lb})$$

where $\mathbf{b} = (\mathbf{X}'\mathbf{X})^{-}\mathbf{X}'\mathbf{y}$ is the estimated solution vector.

The degrees of freedom associated with the hypothesis is equal to the row rank of $\mathbf{L}$. The sum of squares computed in this situation is equivalent to the sum of squares computed using an $\mathbf{L}$ matrix with any row deleted that is a linear combination of previous rows.

Multiple-degree-of-freedom hypotheses can be specified by separating the rows of the $\mathbf{L}$ matrix with commas.

MODEL Statement

MODEL *dependent-variables* = *classification-effects* **;**

The MODEL statement serves two basic purposes.

- The *dependent-variables* specify scenarios for the cell means.
- The *classification-effects* specify the model effects.

Each dependent variable refers to a set of surmised cell means specified in the exemplary data set (which is named by the DATA= option of the PROC GLMPOWER statement). These cell means are response means for all of the subject profiles. Multiple dependent variables correspond to multiple scenarios for these cell means. All models are univariate; the GLMPOWER procedure currently does not support multivariate analyses.

The *classification-effects* specification defines the model effects. You can include main effects and interactions using the effects notation of PROC GLM; see the "Specification of Effects" section on page 1648 in Chapter 31, "The GLM Procedure," for further details.

All variables in the *classification-effects* specification must be contained in the CLASS statement, since power and sample size analyses cover only tests and contrasts of class effects. You can account for covariates in the model by using the NCOVARIATES=, CORRXY=, and PROPVARREDUCTION= options in the POWER statement.

The MODEL statement is required. You can specify only one MODEL statement.

WEIGHT Statement

WEIGHT *variable* **;**

The WEIGHT statement names a variable that provides a profile weight ("cell weight") for each observation in the exemplary data set specified by the DATA= option in the PROC GLMPOWER statement.

If the WEIGHT statement is not used, then a balanced design is assumed with default cell weights of 1.

POWER Statement

POWER < *options* > **;**

The POWER statement performs power and sample size analyses for the test of each effect in the model defined by the MODEL statements and for the contrasts defined by all CONTRAST statements.

Summary of Options

Table 33.4 summarizes categories of options available in the POWER statement.

Table 33.4. Summary of Options in the POWER Statement

Task	Options
Specify significance level	ALPHA=
Specify covariates	CORRXY= NCOVARIATES= PROPVARREDUCTION=
Specify error standard deviation	STDDEV=
Specify sample size	NTOTAL=
Specify power	POWER=
Control sample size rounding	NFRACTIONAL
Control ordering in output	OUTPUTORDER=

Table 33.5 summarizes the valid result parameters.

Table 33.5. Summary of Result Parameters in the POWER Statement

Solve for	Syntax
Power	POWER = .
Sample size	NTOTAL = .

Dictionary of Options

ALPHA=*number-list*

specifies the level of significance of each test. The default is 0.05, corresponding to the usual 0.05 $\times$ 100% = 5% level of significance. Note that this is a test-wise significance level with the same value for all tests, without any corrections for multiple testing. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

OUTPUTORDER=INTERNAL
OUTPUTORDER=REVERSE
OUTPUTORDER=SYNTAX

controls how the input and default analysis parameters are ordered in the output. OUTPUTORDER=INTERNAL (the default) produces the following order.

- weight variable (from the WEIGHT statement)
- source (contrasts from CONTRAST statements, and model effects)
- ALPHA
- dependent variable (from the MODEL statement, representing scenarios for cell means)
- NCOVARIATES

- CORRXY
- PROPVARREDUCTION
- STDDEV
- NTOTAL
- POWER

The OUTPUTORDER=SYNTAX option arranges the parameters in the output in the same order as their corresponding options are specified in the POWER statement. The OUTPUTORDER=REVERSE option arranges the parameters in the output in the reverse order as their corresponding options are specified in the POWER statement.

CORRXY=*number-list*

specifies the multiple correlation ρ between all covariates and the response. The error standard deviation given by the STDDEV= option is consequently reduced by multiplying it by a factor of $(1-\rho^2)^{\frac{1}{2}}$, provided the number of covariates (as determined by the NCOVARIATES= option) is greater than zero. You cannot use the CORRXY= and the PROPVARREDUCTION= options simultaneously. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

NCOVARIATES=*number-list*

specifies the number of covariates to assume as part of the model. This number should be the sum of the number of continuous covariates and the number of levels of all categorical covariates; in other words, the total degrees of freedom represented by covariates. These covariates do not appear in the MODEL statement. The error degrees of freedom are consequently reduced by the number of covariates, and the error standard deviation (whose unadjusted value is provided with the STDDEV= option) may be reduced according to the values supplied by the CORRXY= or PROPVARREDUCTION= options. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

NFRACTIONAL

enables fractional input and output for sample sizes. See the "Sample Size Adjustment Options" section on page 1805 for information on the ramifications of the presence (and absence) of the NFRACTIONAL option.

NTOTAL= *number-list*

specifies the sample size or requests a solution for the sample size with a missing value (NTOTAL=.). Values for the sample size must be no smaller than the model degrees of freedom (counting the covariates). See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

POWER= *number-list*

specifies the desired power of each test or requests a solution for the power with a missing value (POWER=.). The power is expressed as a probability (for example, 0.9) rather than a percentage. Note that this is a test-wise power with the same value for all tests, without any correction for multiple testing. See the "Specifying Value

Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

PROPVARREDUCTION=*number-list*

specifies the proportional reduction (r) in total R^2 incurred by the covariates; in other words, the amount of additional variation explained by the covariates. The error standard deviation given by the STDDEV= option is consequently reduced by multiplying it by a factor of $(1-r)^{\frac{1}{2}}$, provided that the number of covariates (as determined by the NCOVARIATES= option) is greater than zero. You cannot use the PROPVARREDUCTION= and the CORRXY= options simultaneously. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

STDDEV=*number-list*

specifies the error standard deviation, or root MSE. If covariates are specified using the NCOVARIATES= option, then the STDDEV= option denotes the error standard deviation before accounting for these covariates. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

Restrictions on Option Combinations

For the relationship between covariates and response, specify either the multiple correlation (using the CORRXY= option) or the proportional reduction in total R^2 (using the PROPVARREDUCTION= option).

PLOT Statement

PLOT < *plot-options* > < **/** *graph-options* > **;**

The PLOT statement produces a graph or set of graphs for the sample size analysis defined by the previous POWER statement. The *plot-options* define the plot characteristics, and the *graph-options* are SAS/GRAPH-style options.

Options

You can specify the following *plot-options* in the PLOT statement.

INTERPOL=JOIN
INTERPOL=NONE

specifies the type of curve to draw through the computed points. The INTERPOL=JOIN option connects computed points by straight lines. The INTERPOL=NONE option leaves computed points unconnected.

KEY= BYCURVE < **(** *bycurve-options* **)** >
KEY= BYFEATURE < **(** *byfeature-options* **)** >
KEY= ONCURVES

specifies the style of key (or "legend") for the plot. The default is KEY=BYFEATURE, which specifies a key with a column of entries for each used plot feature (line style, color, and/or symbol). Each entry shows the mapping between a value of the feature and the value(s) of the analysis parameter(s) linked to that feature. The KEY=BYCURVE option specifies a key with each row identifying a distinct curve in the plot. The KEY=ONCURVES option places a curve-specific label adjacent to each curve.

You can specify the following *byfeature-options* in parentheses after the KEY=BYCURVE option.

NUMBERS=OFF

NUMBERS=ON specifies how the key should identify curves. If NUMBERS=OFF, then the key includes symbol, color, and line style samples to identify the curves. If NUMBERS=ON, then the key includes numbers matching numeric labels placed adjacent to the curves. The default is NUMBERS=ON.

POS=BOTTOM

POS=INSET specifies the position of the key. The POS=BOTTOM option places the key below the x-axis. The POS=INSET option places the key inside the plotting region and attempts to choose the least crowded corner. The default is POS=BOTTOM.

You can specify the following *byfeature-options* in parentheses after KEY=BYFEATURE option.

POS=BOTTOM

POS=INSET specifies the position of the key. The POS=BOTTOM option places the key below the x-axis. The POS=INSET option places the key inside the plotting region and attempts to choose the least crowded corner. The default is POS=BOTTOM.

MARKERS=ANALYSIS
MARKERS=COMPUTED
MARKERS=NICE
MARKERS=NONE

specifies the locations for plotting symbols.

The MARKERS=ANALYSIS option places plotting symbols at locations corresponding to the values of the parameter associated with the “argument” axis (the axis that is *not* representing the parameter being solved for) from the POWER statement preceding the PLOT statement.

The MARKERS=COMPUTED option (the default) places plotting symbols at the locations of actual computed points from the sample size analysis (as opposed to interpolated values).

The MARKERS=NICE option places plotting symbols at tick mark locations (corresponding to the argument axis).

The MARKERS=NONE option disables plotting symbols.

MAX=*number*

specifies the maximum of the range of values for the parameter associated with the "argument" axis (the axis that is *not* representing the parameter being solved for). The default is the maximum value occurring for this parameter in the POWER statement preceding the PLOT statement.

MIN=*number*

specifies the minimum of the range of values for the parameter associated with the "argument" axis (the axis that is *not* representing the parameter being solved for). The default is the minimum value occurring for this parameter in the POWER statement preceding the PLOT statement.

NPOINTS=*number*

specifies the number of values for the parameter associated with the "argument" axis (the axis that is *not* representing the parameter being solved for). You cannot use the NPOINTS= and STEP= options simultaneously. The default value for typical situations is 20.

STEP=*number*

specifies the increment between values of the parameter associated with the "argument" axis (the axis that is *not* representing the parameter being solved for). You cannot use the STEP= and NPOINTS= options simultaneously. By default, the NPOINTS= option is used instead of the STEP= option.

VARY (*feature* < **BY** *parameter-list* > **...** *feature* < **BY** *parameter-list* > **)**

specifies how plot features should be linked to varying analysis parameters. Available plot *features* are COLOR, LINESTYLE, PANEL, and SYMBOL. A "panel" refers to a separate plot with a heading identifying the subset of values represented in the plot.

The *parameter-list* is a list of one or more names separated by spaces. Each name must match the name of an analysis option used in the POWER statement preceding the PLOT statement, *or* one of the following keywords: SOURCE (for the tests) and DEPENDENT (for the cell means scenarios).

If you omit the < BY *parameter-list* > portion for a feature, then one or more multi-valued parameters from the analysis will be automatically selected for you.

X=N
X=POWER

specifies a plot with the requested type of parameter on the x-axis and the parameter being solved for on the y-axis. When X=N, sample size is assigned to the x-axis. When X=POWER, power is assigned to the x-axis. You cannot use the X= and Y= options simultaneously. The default is X=POWER, unless the result parameter is power, in which case the default is X=N.

XOPTS= (*x-options* **)**

specifies plot characteristics pertaining to the x-axis.

You can specify the following *x-options* in parentheses.

CROSSREF=NO
CROSSREF=YES specifies whether the reference lines defined by the REF= *x-option* should turn at the intersection with each curve and extend to the y-axis.

REF=*number-list* specifies locations for reference lines extending from the x-axis across the entire plotting region. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

Y=N
Y=POWER
specifies a plot with the requested type of parameter on the y-axis and the parameter being solved for on the x-axis. When Y=N, sample size is assigned to the y-axis. When Y=POWER, power is assigned to the y-axis. You cannot use the Y= and X= options simultaneously. By default, the X= option is used instead of the Y= option.

YOPTS= (*y-options* **)**
specifies plot characteristics pertaining to the y-axis.

You can specify the following *y-options* in parentheses.

CROSSREF=NO

CROSSREF=YES specifies whether the reference lines defined by the REF= *y-option* should turn at the intersection with each curve and extend to the x-axis.

REF=*number-list* specifies locations for reference lines extending from the y-axis across the entire plotting region. See the "Specifying Value Lists in the POWER Statement" section on page 1805 for information on specifying the *number-list*.

You can specify the following *graph-options* in the PLOT statement after a slash (/).

DESCRIPTION='*string* **'**
specifies a descriptive string of up to 40 characters that appears in the "Description" field of the graphics catalog. The description does not appear on the plots. By default, PROC GLMPOWER assigns a description either of the form "*Y* versus *X*" (for a single-panel plot) or of the form "*Y* versus *X* (*S*)," where *Y* is the parameter on the y-axis, *X* is the parameter on the x-axis, and *S* is a description of the subset represented on the current panel of a multipanel plot.

NAME='*string* **'**
specifies a name of up to eight characters for the catalog entry for the plot. The default name is PLOT*n*, where *n* is the number of the plot statement within the current invocation of PROC GLMPOWER. If the name duplicates the name of an existing entry, SAS/GRAPH software adds a number to the duplicate name to create a unique entry, for example, PLOT11 and PLOT12 for the second and third panels of a multipanel plot generated in the first PLOT statement in an invocation of PROC GLMPOWER.

Details

Specifying Value Lists in the POWER Statement

To specify one or more scenarios for an analysis parameter (or set of parameters) in the POWER statement, you provide a list of values attached to the option representing the parameter(s). To identify the parameter you wish to solve for, you place missing values in the appropriate list.

Scenarios for scalar-valued parameters, such as power, are represented by a *number-list*.

Number-lists

A *number-list* can be one of two things: a series of one or more numbers expressed in the form of one or more DOLISTs, or a missing value indicator (.).

The DOLIST format is the same as in the DATA step language. For example, you can specify four scenarios (30, 50, 70, and 100) for a total sample size in any of the following ways.

```
NTOTAL = 30 50 70 100
NTOTAL = 30 to 70 by 20 100
```

A missing value identifies a parameter as the result parameter; it is valid only with options representing parameters you can solve for in a given analysis. For example, you can request a solution for NTOTAL:

```
NTOTAL = .
```

Sample Size Adjustment Options

Without the NFRACTIONAL option, sample sizes are rounded conservatively (down in the input, up in the output) so that all total sizes *and* sample sizes for individual design profiles are integers. In addition, all design profile sizes are adjusted to be multiples of their corresponding weights. If a design profile is present more than once in the exemplary data set, then the weights for that design profile are summed. For example, if a particular design profile is present twice in the exemplary data set with weight values 2 and 6, then all sample sizes for this design profile become multiples of $2 + 6 = 8$.

With the NFRACTIONAL option, sample size input is left alone, and sample size results are reported in two versions, a raw “fractional” version and a “ceiling” version rounded up to the nearest integer.

Whenever an input sample size is adjusted, both the original (“nominal”) and adjusted (“actual”) sample sizes are reported. Whenever computed sample sizes are adjusted, both the original input (“nominal”) power and the achieved (“actual”) power at the adjusted sample size are reported.

Error and Information Output

The Error column in the main output table explains reasons for missing results and flags numerical results that are bounds rather than exact answers.

The Information column provides further details about Error entries, warnings about any boundary conditions detected, and notes about any adjustments to input. Note that the Information column is hidden by default in the main output. But you can view it as the Info column in the Output data set by using the ODS OUTPUT statement and the PRINT procedure. For example, the following SAS statements print both the Error and Info columns for a power computation in a one-way ANOVA.

```
   data MyExemp;
      input A $ Y1 Y2;
      datalines;
            1   10 11
            2   12 11
            3   15 11
      ;
   run;


proc glmpower data=MyExemp;
   class A;
   model Y1 Y2 = A;
   power
      stddev = 2
      ntotal = 3 10
      power  = .;
   ods output output=Power;
proc print data=Power;
   var NominalNTotal NTotal Dependent Power Error Info;
run;
```

The output is shown in Figure 33.5.

```
    Nominal
Obs NTotal  NTotal Dependent    Power     Error      Info

 1      3      3      Y1      .        Invalid input Error DF=0
 2     10      9      Y1     0.557156               Input N adjusted
 3      3      3      Y2      .        Invalid input Error DF=0 / No effect
 4     10      9      Y2     0.050000               Input N adjusted / No effect
```

Figure 33.5. Error and Information Columns

The sample size of 3 specified with the NTOTAL= option causes an “Invalid input” message in the Error column and an “Error DF=0” message in the Info column, because a sample size of 3 is so small that there are no degrees of freedom left for the error term. The sample size of 10 causes an “Input N adjusted” message in the Info column because it is rounded down to 9 to produce integer group sizes of 3 per cell.

The cell means scenario represented by the dependent variable Y2 causes a "No effect" message to appear in the Info column, because the means in this scenario are all equal.

Displayed Output

If you use the PLOTONLY option in the PROC GLMPOWER statement, the procedure only displays graphical output. Otherwise, the displayed output of the GLMPOWER procedure includes the following.

- the description of the statistical analysis
- the "Fixed Scenario Elements" table, which shows all applicable single-valued analysis parameters, in the following order: the weight variable, the source of the test, parameters input explicitly, parameters supplied with defaults, and ancillary results
- an output table showing the following when applicable (in order): the index of the scenario, the source of the test, all multivalued input, ancillary results, the primary computed result, and error descriptions
- plots (if requested)

Ancillary results include the following.

- Actual Power, the achieved power, if it differs from the input (Nominal) power value
- fractional sample size, if the NFRACTIONAL option is used in the analysis statement

If sample size is the result parameter and the NFRACTIONAL option is used in the analysis statement, then both "Fractional" and "Ceiling" sample size results are displayed. Fractional sample sizes correspond to the "Nominal" values of power or precision probability. Ceiling sample sizes are simply the fractional sample sizes rounded up to the nearest integer; they correspond to "Actual" values of power or precision probability.

ODS Table Names

PROC GLMPOWER assigns a name to each table that it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in Table 33.6. For more information on ODS, see Chapter 14, "Using the Output Delivery System."

Table 33.6. ODS Tables Produced in PROC GLMPOWER

ODS Table Name	Description	Statement
FixedElements	factoid with single-valued analysis parameters	default

Table 33.6. (continued)

ODS Table Name	Description	Statement
Output	all input and computed analysis parameters, error messages, and information messages for each scenario	default
PlotContent	data contained in plots, including analysis parameters and indices identifying plot features	PLOT

The ODS path names are created as follows.

- Glmpower.Power<n>.FixedElements
- Glmpower.Power<n>.Output
- Glmpower.Power<n>.PlotContent
- Glmpower.Power<n>.Plot<n>

where

- The Plot<n> objects are the actual plots.
- The <n> indexing the Power statement is only used if there is more than one instance.
- The <n> indexing the plots increases with every panel in every plot statement, resetting to 1 only at new analysis statements.

Mathematical Methods and Formulas

This section describes the approaches used in PROC GLMPOWER to compute power for each analysis. Unless otherwise indicated, computed values for parameters besides power (for example, sample size) are obtained by solving power formulas for the desired parameters.

Contrasts in Fixed-Effect Univariate Models

The univariate linear model is

$$\mathbf{y} = \mathbf{X}\boldsymbol{\beta} + \boldsymbol{\epsilon}$$

where $\mathbf{y}$ is the $N \times 1$ vector of responses, $\mathbf{X}$ is the $N \times p_x$ design matrix, $\boldsymbol{\beta}$ is the $p_x \times 1$ vector of model parameters corresponding to the columns of $\mathbf{X}$, and $\boldsymbol{\epsilon}$ is an $N \times 1$ vector of errors with

$$\epsilon_1, \ldots, \epsilon_N \sim \mathrm{N}(0, \sigma^2) \quad \text{(i.i.d.)}$$

The general linear hypothesis for the univariate model is

$$\begin{aligned} H_0 &: \mathbf{L}\boldsymbol{\beta} = \theta_{\mathbf{0}} \\ H_A &: \mathbf{L}\boldsymbol{\beta} \neq \theta_{\mathbf{0}} \end{aligned}$$

where $\mathbf{L}$ is an $r_L \times p_x$ contrast matrix (assumed to be full rank), and $\theta_{\mathbf{0}}$ is the null value (usually just a vector of zeroes). Note that effect tests are just contrasts using special forms of $\mathbf{L}$. Thus, this scheme covers both effect tests and custom contrasts.

The test statistic is the following:

$$F = \frac{\left(\frac{\mathrm{SSH}}{r_L}\right)}{\hat{\sigma}^2}$$

where

$$\begin{aligned} \mathrm{SSH} &= \frac{1}{N}\left(\mathbf{L}\hat{\boldsymbol{\beta}} - \theta_{\mathbf{0}}\right)' \left(\mathbf{L}\left(\mathbf{X}'\mathbf{X}\right)^{-1}\mathbf{L}'\right)^{-1}\left(\mathbf{L}\hat{\boldsymbol{\beta}} - \theta_{\mathbf{0}}\right) \\ \hat{\boldsymbol{\beta}} &= (\mathbf{X}'\mathbf{X})^{-1}\mathbf{X}'\mathbf{y} \\ \hat{\sigma}^2 &= \frac{1}{N - r_x}\left(\mathbf{y} - \mathbf{X}\hat{\boldsymbol{\beta}}\right)'\left(\mathbf{y} - \mathbf{X}\hat{\boldsymbol{\beta}}\right) \end{aligned}$$

where r_x is the rank of $\mathbf{X}$ (and equal to p_x if $\mathbf{X}$ is full-rank).

Under H_0, $F \sim F(r_L, N - r_x)$.

In PROC GLMPOWER, the model parameters $\boldsymbol{\beta}$ are not specified directly, but rather indirectly as $\mathbf{y}^\star$, which represents either conjectured response means or typical response values for each design profile. The $\mathbf{y}^\star$ values are manifested as the dependent variable in the MODEL statement. The vector $\boldsymbol{\beta}$ is obtained from $\mathbf{y}^\star$ according to the least squares equation,

$$\boldsymbol{\beta} = (\mathbf{X}'\mathbf{X})^{-1}\mathbf{X}'\mathbf{y}^\star$$

Note that, in general, there is not a 1-1 mapping between $\mathbf{y}^\star$ and $\boldsymbol{\beta}$. Many different scenarios for $\mathbf{y}^\star$ may lead to the same $\boldsymbol{\beta}$. If you specify $\mathbf{y}^\star$ with the intention of representing cell means, keep in mind that PROC GLMPOWER will allow scenarios that are *not* valid cell means according to the model specified in the MODEL statement. For example, if $\mathbf{y}^\star$ possesses an interaction effect but the corresponding interaction term is left out of the model, then the cell means ($\mathbf{X}\boldsymbol{\beta}$) derived from $\boldsymbol{\beta}$ will differ from $\mathbf{y}^\star$. In particular, the cell means thus derived are the projection of $\mathbf{y}^\star$ onto the model space.

It is convenient in power analysis to parameterize the design matrix $\mathbf{X}$ in three parts, $\{\ddot{\mathbf{X}}, \mathbf{w}, N\}$, defined as follows.

1. The $q \times p_x$ essence design matrix $\ddot{\mathbf{X}}$ is the collection of unique rows of $\mathbf{X}$. Its rows are sometimes referred to as "design profiles." Here, $q \leq N$ is defined simply as the number of unique rows of $\mathbf{X}$.

2. The $q \times 1$ weight vector $\mathbf{w}$ reveals the relative proportions of design profiles. Row i of $\ddot{\mathbf{X}}$ is to be included in the design w_i times for every w_j times row j is included. The weights are assumed to be standardized (i.e., sum up to 1).
3. The total sample size is N. This is the number of rows in $\mathbf{X}$. If you gather $Nw_i = n_i$ copies of the i^{th} row of $\ddot{\mathbf{X}}$, for $i = 1, \ldots, q$, then you end up with $\mathbf{X}$.

It is useful to express the the cross-product matrix $\mathbf{X}'\mathbf{X}$ in terms of these three parts,

$$\mathbf{X}'\mathbf{X} = N\ddot{\mathbf{X}}'\text{diag}(\mathbf{w})\ddot{\mathbf{X}}$$

since this factors out the portion (N) depending on sample size and the portion ($\ddot{\mathbf{X}}'\text{diag}(\mathbf{w})\ddot{\mathbf{X}}$) depending only on the design structure.

Under H_A, the test statistic F is distributed as $F(r_L, N - r_x, \lambda)$ with noncentrality

$$\lambda = N\left(\mathbf{L}\boldsymbol{\beta} - \theta_{\mathbf{0}}\right)' \left(\mathbf{L}\left(\ddot{\mathbf{X}}'\text{diag}(\mathbf{w})\ddot{\mathbf{X}}\right)^{-1}\mathbf{L}'\right)^{-1} \left(\mathbf{L}\boldsymbol{\beta} - \theta_{\mathbf{0}}\right)\sigma^{-2}$$

Muller and Peterson (1984) give the exact power of the test as

$$\text{power} = P\left(F(r_L, N - r_x, \lambda) \geq F_{1-\alpha}(r_L, N - r_x)\right)$$

Refer to Muller et al. (1992) and O'Brien and Shieh (1992) for additional discussion.

Adjustments for Covariates

If you specify covariates in the model (whether continuous or categorical), then two adjustments are made in order to compute approximate power in the presence of the covariates. Let n_ν denote the number of covariates (counting dummy variables for categorical covariates individually). In other words, n_ν is the total degrees of freedom used by the covariates. The adjustments are the following.

1. The error degrees of freedom decreases by n_ν.
2. The error standard deviation σ shrinks by a factor of $(1-\rho^2)^{\frac{1}{2}}$ (if the CORRXY= option to specify the correlation ρ between covariates and response) or $(1-r)^{\frac{1}{2}}$ (if the PROPVARREDUCTION= option is used to specify the proportional reduction in total R^2 incurred by the covariates). Let $\sigma^\star$ represent the updated value of σ.

As a result of these changes, the power is computed as

$$\text{power} = P\left(F(r_L, N - r_x - n_\nu, \lambda^\star) \geq F_{1-\alpha}(r_L, N - r_x - n_\nu)\right)$$

where $\lambda^\star$ is calculated using $\sigma^\star$ rather than σ:

$$\lambda^\star = N\left(\mathbf{L}\boldsymbol{\beta} - \theta_{\mathbf{0}}\right)' \left(\mathbf{L}\left(\ddot{\mathbf{X}}'\text{diag}(\mathbf{w})\ddot{\mathbf{X}}\right)^{-1}\mathbf{L}'\right)^{-1} \left(\mathbf{L}\boldsymbol{\beta} - \theta_{\mathbf{0}}\right)(\sigma^\star)^{-2}$$

Examples

Example 33.1. One-Way ANOVA

This example duplicates Example 56.1 on page 3243 of Chapter 56, "The POWER Procedure," using PROC GLMPOWER to perform the same sample size analysis.

Hocking (1985, p. 109) describes a study of the effectiveness of electrolytes in reducing lactic acid build-up for long-distance runners. You are planning a similar study in which you will allocate five different fluids to runners on a 10-mile course and measure lactic acid build-up immediately after the race. The fluids consist of water and two commercial electrolyte drinks, A and B, each prepared at two concentrations, low (A1 and B1) and high (A2 and B2).

You conjecture that the standard deviation of lactic acid measurements given any particular fluid is about 3.75, and that the expected lactic acid values will correspond roughly to Table 33.7. You are least familiar with the B1 drink and hence decide to consider reasonable upper and lower values for that mean.

Table 33.7. Mean Lactic Acid Build-up by Fluid

Water	A1	A2	B1	B2
35.6	33.7	30.2	29 or 28	25.9

You are interested in four different comparisons, shown in Table 33.8 with appropriate contrast coefficients.

Table 33.8. Planned Comparisons

	Contrast Coefficients				
Comparison	Water	A1	A2	B1	B2
Water versus electrolytes	4	-1	-1	-1	-1
A versus B	0	1	1	-1	-1
A1 versus A2	0	1	-1	0	0
B1 versus B2	0	0	0	1	-1

For each of these contrasts you want to determine the sample size required to achieve a power of 0.9 to detect an effect with magnitude according to Table 33.7. You are not yet attempting to choose a single sample size for the study, but rather checking the range of sample sizes needed by individual contrasts. You plan to test each contrast at $\alpha = 0.025$. You will provide twice as many runners with water as with any of the electrolytes; that is, you will use a sample size weighting scheme of 2:1:1:1:1.

Before calling PROC GLMPOWER, you need to create the *exemplary data set* to specify means and weights for the design profiles:

```
data Fluids;
   input Fluid $ LacticAcid1 LacticAcid2 CellWgt;
   datalines;
         Water       35.6        35.6         2
         A1          33.7        33.7         1
         A2          30.2        30.2         1
         B1          29          28           1
```

```
            B2          25.9         25.9         1
         ;
      run;
```

The variable LacticAcid1 represents the cell means scenario with the larger B1 mean (29), and LacticAcid2 represents the scenario with the smaller B1 mean (28). The variable CellWgt contains the sample size allocation weights.

Use the DATA= option in the PROC GLMPOWER statement to specify Fluids as the exemplary data set. Identify Fluid as a classification variable with the CLASS statement. Specify the model using the MODEL statement, including both cell means scenarios LacticAcid1 and LacticAcid2 as dependent variables. Identify CellWgt as the weight variable using the WEIGHT statement. Specify the contrasts with the CONTRAST statement. In doing so, keep in mind that PROC GLMPOWER processes class levels in order of formatted values. Thus you need to order the contrast coefficients to correspond to the following order: A1, A2, B1, B2, Water. Use the POWER statement to specify total sample size as the result parameter and provide values for the other analysis parameters (error standard deviation, alpha, and power).

```
proc glmpower data=Fluids;
   class Fluid;
   model LacticAcid1 LacticAcid2 = Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
   contrast "B1 vs. B2"              Fluid   0  0  1 -1 0;
   power
      stddev = 3.75
      alpha  = 0.025
      ntotal = .
      power  = 0.9;
run;
```

The NTOTAL= option in the POWER statement identifies total sample size as the result parameter with a missing value (.). The STDDEV= option specifies an error standard deviation of 3.75; the ALPHA= option specifies a significance level of 0.025; and the POWER= option specifies a target power of 0.9.

Output 33.1.1 displays the results.

Output 33.1.1. Sample Sizes for One-Way ANOVA Contrasts

```
                         The GLMPOWER Procedure
     Tests of contrasts and effects in the fixed-effects general linear model

                        Fixed Scenario Elements

                Weight Variable                  CellWgt
                Alpha                              0.025
                Error Standard Deviation            3.75
                Nominal Power                        0.9


                            Computed N Total

                                              Test   Error    Actual       N
      Index          Source          Dependent   DF      DF     Power   Total

          1  Water vs. electrolytes  LacticAcid1     1      25  0.946500      30
          2  Water vs. electrolytes  LacticAcid2     1      19  0.901344      24
          3  A vs. B                 LacticAcid1     1      55  0.928644      60
          4  A vs. B                 LacticAcid2     1      43  0.921942      48
          5  A1 vs. A2               LacticAcid1     1     169  0.900801     174
          6  A1 vs. A2               LacticAcid2     1     169  0.900801     174
          7  B1 vs. B2               LacticAcid1     1     217  0.902098     222
          8  B1 vs. B2               LacticAcid2     1     475  0.901643     480
          9  Fluid                   LacticAcid1     4      25  0.958477      30
         10  Fluid                   LacticAcid2     4      25  0.971701      30
```

The sample sizes range from 24 for the comparison of water versus electrolytes to 480 for the comparison of B1 versus B2, both assuming the smaller B1 mean. The sample size for this latter comparison is relatively large because the small mean difference of $28 - 25.9 = 2.1$ is hard to detect. PROC GLMPOWER also includes the effect test for Fluid. Note that, in this case, it is equivalent to TEST=OVERALL_F in the ONEWAYANOVA statement of PROC POWER, since there is only one effect in the model.

The Nominal Power of 0.9 in the Fixed Scenario Elements table in Output 33.1.1 represents the input target power, and the Actual Power column in the Computed N Total table is the power at the sample size (N Total) adjusted to achieve the specified sample weighting exactly. Note that all of the sample sizes are rounded up to multiples of 6 to preserve integer group sizes (since the group weights add up to 6). You can use the NFRACTIONAL option in the POWER statement to compute raw fractional sample sizes.

Suppose you want to plot the required sample size for the range of power values from 0.5 to 0.95. First, define the analysis by specifying the same statements as before, but add the PLOTONLY option to the PROC GLMPOWER statement to disable the non-graphical results. Next, specify the PLOT statement with X=POWER to request a plot with power on the x-axis. (The result parameter, here sample size, is always plotted on the other axis). Use the MIN= and MAX= options in the PLOT statement to specify the power range.

```
proc glmpower data=Fluids plotonly;
   class Fluid;
   model LacticAcid1 LacticAcid2 = Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
   contrast "B1 vs. B2"              Fluid   0  0  1 -1 0;
   power
      stddev = 3.75
      alpha  = 0.025
      ntotal = .
      power  = 0.9;
   plot x=power min=.5 max=.95;
run;
```

See Output 33.1.2 for the resulting plot.

Output 33.1.2. Plot of Sample Size versus Power for One-Way ANOVA Contrasts



In Output 33.1.2, the line style identifies the test, and the plotting symbol identifies the cell means scenario. The plotting symbol locations identify actual computed powers; the curves are linear interpolations of these points. The plot shows that the required sample size is highest for the test of B1 versus B2, for either cell means scenario.

Note that some of the plotted points in Output 33.1.2 are unevenly spaced. This is because the plotted points are the *rounded* sample size results at their corresponding *actual* power levels. The range specified with the MIN= and MAX= values in the PLOT statement correspond to *nominal* power levels. In some cases, actual power is substantially higher than nominal power. To obtain plots with evenly spaced points (but with *fractional* sample sizes at the computed points), you can use the NFRACTIONAL option in the POWER statement preceding the PLOT statement.

Finally, suppose you want to plot the power for the range of sample sizes you will likely consider for the study (the range of 24 to 480 that achieves 0.9 power for different comparisons). Define the analysis in PROC GLMPOWER by specifying the DATA= and PLOTONLY options on the PROC GLMPOWER statement, the MODEL statement, the WEIGHT statement, and the CONTRAST statements as before. In the POWER statement, identify power as the result (POWER=.), and specify NTOTAL=24. Use the STDDEV= and ALPHA= options as before. Specify the PLOT statement with X=N to request a plot with sample size on the x-axis. Use the MIN= and MAX= options in the PLOT statement to specify the sample size range.

```
proc glmpower data=Fluids plotonly;
   class Fluid;
   model LacticAcid1 LacticAcid2 = Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
   contrast "B1 vs. B2"              Fluid   0  0  1 -1 0;
   power
      stddev = 3.75
      alpha  = 0.025
      ntotal = 24
      power  = .;
   plot x=n min=24 max=480;
run;
```

Note that the value specified with the NTOTAL= option (24) is not used. It is overridden in the plot by the MIN= and MAX= options in the PLOT statement, and the PLOTONLY option in the PROC GLMPOWER statement disables non-graphical results. But the NTOTAL= option (along with a value) is still needed in the POWER statement as a placeholder, to identify the desired parameterization for sample size.

See Output 33.1.3 for the plot.

Output 33.1.3. Plot of Power versus Sample Size for One-Way ANOVA Contrasts



Although Output 33.1.2 and Output 33.1.3 surface essentially the same computations for practical power ranges, they each provide a different quick visual assessment. Output 33.1.2 reveals the range of required sample sizes for powers of interest, and Output 33.1.3 reveals the range of powers achieved for sample sizes of interest.

Example 33.2. Two-Way ANOVA with Covariate

Suppose you can enhance the planned study discussed in Example 33.1 on page 1811 in two ways:

- Incorporate results from races at two different altitudes (“high” and “low”).
- Measure the body mass of each runner before the race.

This is equivalent to adding a second fixed effect and a continuous covariate to your model.

Since lactic acid build-up is more pronounced at higher altitudes, you will include altitude as a factor in the model along with fluid, extending the one-way ANOVA to a two-way ANOVA. In doing so, you expect to lower the residual standard deviation from about 3.75 to 3.5 (in addition to generalizing the study results). You assume there is negligible interaction between fluid and altitude and plan to use a main-effects-only model. You conjecture that the mean lactic acid build-up follows Table 33.9.

Table 33.9. Mean Lactic Acid Build-up by Fluid and Altitude

	Fluid				
Altitude	Water	A1	A2	B1	B2
High	36.9	35.0	31.5	30	27.1
Low	34.3	32.4	28.9	27	24.7

By including a measurement of body mass as a covariate in the study, you hope to further reduce the error variability. The extent of this reduction in variability is commonly expressed in two alternative ways: (1) the correlation between the covariates and the response or (2) the proportional reduction in total R^2 incurred by the covariates. You prefer the former and guess that the correlation between body mass and lactic acid build-up is between 0.2 and 0.3. You specify these estimates with the NCOVARIATES= and CORRXY= options in the POWER statement. The covariate is not included in the MODEL statement.

You are interested in the same four fluid comparisons as in Example 33.1, shown in Table 33.8 on page 1811, except this time you want to marginalize over the effect of altitude.

For each of these contrasts, you want to determine the sample size required to achieve a power of 0.9 to detect an effect with magnitude according to Table 33.9. You are not yet attempting to choose a single sample size for the study, but rather checking the range of sample sizes needed by individual contrasts. You plan to test each contrast at $\alpha = 0.025$. You will provide twice as many runners with water as with any of the electrolytes, and you predict that you can study approximately 2/3 as many runners at the high altitude than at the low altitude. The resulting planned sample size weighting scheme is shown in Table 33.10. Since the scheme is only approximate, you use the NFRACTIONAL option in the POWER statement to disable the rounding of sample sizes up to integers satisfying the weights exactly.

Table 33.10. Approximate Sample Size Allocation Weights

	Fluid				
Altitude	Water	A1	A2	B1	B2
High	4	2	2	2	2
Low	6	3	3	3	3

First, you create the exemplary data set to specify means and weights for the design profiles:

```
data Fluids2;
   input Altitude $ Fluid $ LacticAcid CellWgt;
   datalines;
         High      Water      36.9      4
         High      A1         35.0      2
         High      A2         31.5      2
         High      B1         30        2
         High      B2         27.1      2
         Low       Water      34.3      6
         Low       A1         32.4      3
         Low       A2         28.9      3
         Low       B1         27        3
```

```
            Low        B2          24.7       3
      ;
   run;
```

The variables Altitude, Fluid, and LacticAcid specify the factors and cell means in Table 33.9. The variable CellWgt contains the sample size allocation weights in Table 33.10.

Use the DATA= option in the PROC GLMPOWER statement to specify Fluids2 as the exemplary data set. Identify Altitude and Fluid classification variables with the CLASS statement. Specify the model using the MODEL statement, and identify CellWgt as the weight variable using the WEIGHT statement. Specify the contrasts in Table 33.8 on page 1811 with the CONTRAST statement. As in Example 33.1, order the contrast coefficients in order of formatted class levels (A1, A2, B1, B2, Water). Use the POWER statement to specify total sample size as the result parameter and provide values for the other analysis parameters (error standard deviation, number of covariates, correlation between covariate and response, alpha, and power).

```
proc glmpower data=Fluids2;
   class Altitude Fluid;
   model LacticAcid = Altitude Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
   contrast "B1 vs. B2"              Fluid   0  0  1 -1 0;
   power
      nfractional
      stddev      = 3.5
      ncovariates = 1
      corrxy      = 0.2 0.3
      alpha       = 0.025
      ntotal      = .
      power       = 0.9;
run;
```

The NCOVARIATES= option specifies the single covariate (body mass), and the CORRXY= option specifies the two scenarios for its correlation with lactic acid build-up (0.2 and 0.3).

Output 33.2.1 displays the results.

Output 33.2.1. Sample Sizes for Two-Way ANOVA Contrasts

The GLMPOWER Procedure
Tests of contrasts and effects in the fixed-effects general linear model

Fixed Scenario Elements

Weight Variable	CellWgt
Alpha	0.025
Dependent Variable	LacticAcid
Number of Covariates	1
Std Dev Without Covariate Adj	3.5
Nominal Power	0.9

Computed Ceiling N Total

Index	Source	Corr XY	Adj Std Dev	Test DF	Adj Error DF	Fractional N Total
1	Water vs. electrolytes	0.2	3.43	1	15	21.720195
2	Water vs. electrolytes	0.3	3.34	1	14	20.848805
3	A vs. B	0.2	3.43	1	35	41.657424
4	A vs. B	0.3	3.34	1	33	39.674037
5	A1 vs. A2	0.2	3.43	1	139	145.613657
6	A1 vs. A2	0.3	3.34	1	132	138.173983
7	B1 vs. B2	0.2	3.43	1	268	274.055008
8	B1 vs. B2	0.3	3.34	1	253	259.919126
9	Altitude	0.2	3.43	1	84	90.418451
10	Altitude	0.3	3.34	1	79	85.862649
11	Fluid	0.2	3.43	4	16	22.446173
12	Fluid	0.3	3.34	4	15	21.687544

Computed Ceiling N Total

Index	Source	Actual Power	Ceiling N Total
1	Water vs. electrolytes	0.904987	22
2	Water vs. electrolytes	0.902893	21
3	A vs. B	0.902740	42
4	A vs. B	0.902755	40
5	A1 vs. A2	0.900833	146
6	A1 vs. A2	0.901870	139
7	B1 vs. B2	0.901071	275
8	B1 vs. B2	0.900097	260
9	Altitude	0.902036	91
10	Altitude	0.900511	86
11	Fluid	0.912294	23
12	Fluid	0.907508	22

The sample sizes in Output 33.2.1 range from 21 for the comparison of water versus electrolytes (assuming a correlation of 0.3 between body mass and lactic acid build-up) to 275 for the comparison of B1 versus B2 (assuming a correlation of 0.2). PROC GLMPOWER also includes the effect tests for Altitude and Fluid. Note that the required sample sizes for this study are lower than those for the study in Example 33.1.

Note that the error standard deviation has been reduced from 3.5 to 3.43 (when correlation is 0.2) or 3.34 (when correlation is 0.3) in the approximation of the effect of the body mass covariate. The error degrees of freedom has also been automatically adjusted, lowered by 1 (the number of covariates).

Suppose you want to plot the required sample size for the range of power values from 0.5 to 0.95. First, define the analysis by specifying the same statements as before, but add the PLOTONLY option to the PROC GLMPOWER statement to disable the non-graphical results. Next, specify the PLOT statement with X=POWER to request a plot with power on the x-axis. Sample size is automatically placed on the y-axis. Use the MIN= and MAX= options in the PLOT statement to specify the power range.

```
proc glmpower data=Fluids2 plotonly;
   class Altitude Fluid;
   model LacticAcid = Altitude Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
   contrast "B1 vs. B2"              Fluid   0  0  1 -1 0;
   power
      nfractional
      stddev      = 3.5
      ncovariates = 1
      corrxy      = 0.2 0.3
      alpha       = 0.025
      ntotal      = .
      power       = 0.9;
   plot x=power min=.5 max=.95;
run;
```

See Output 33.2.2 for the plot.

Output 33.2.2. Plot of Sample Size versus Power for Two-Way ANOVA Contrasts



In Output 33.1.2, the line style identifies the test, and the plotting symbol identifies the scenario for the correlation between covariate and response. The plotting symbol locations identify actual computed powers; the curves are linear interpolations of these points. As in Example 33.1, the required sample size is highest for the test of B1 versus B2.

Finally, suppose you want to plot the power for the range of sample sizes you will likely consider for the study (the range of 21 to 275 that achieves 0.9 power for different comparisons). Define the analysis in PROC GLMPOWER by specifying the DATA= and PLOTONLY options on the PROC GLMPOWER statement, the MODEL statement, the WEIGHT statement, and the CONTRAST statements as before. In the POWER statement, identify power as the result (POWER=.), and specify NTOTAL=21. Specify the PLOT statement with X=N to request a plot with sample size on the x-axis. Set the MAX= option in the PLOT statement to 275. The MIN= option will automatically default to the value of 21 from the NTOTAL= option in the POWER statement.

```
proc glmpower data=Fluids2 plotonly;
   class Altitude Fluid;
   model LacticAcid = Altitude Fluid;
   weight CellWgt;
   contrast "Water vs. electrolytes" Fluid  -1 -1 -1 -1 4;
   contrast "A vs. B"                Fluid   1  1 -1 -1 0;
   contrast "A1 vs. A2"              Fluid   1 -1  0  0 0;
```

```
      contrast "B1 vs. B2"           Fluid   0  0  1 -1 0;
      power
         nfractional
         stddev      = 3.5
         ncovariates = 1
         corrxy      = 0.2 0.3
         alpha       = 0.025
         ntotal      = 21
         power       = .;
      plot x=n min=21 max=275;
   run;
```

See Output 33.2.3 for the plot.

Output 33.2.3. Plot of Power versus Sample Size for Two-Way ANOVA Contrasts



Although Output 33.2.2 and Output 33.2.3 surface essentially the same computations for practical power ranges, they each provide a different quick visual assessment. Output 33.2.2 reveals the range of required sample sizes for powers of interest, and Output 33.2.3 reveals the range of powers achieved for sample sizes of interest.

References

Castelloe, J.M. (2000), "Sample Size Computations and Power Analysis with the SAS ® System," *Proceedings of the Twenty-Fifth Annual SAS Users Group International Conference,* Paper 265-25, Cary, NC: SAS Institute Inc.

Castelloe, J.M. and O'Brien, R.G. (2001), "Power and Sample Size Determination for Linear Models," *Proceedings of the Twenty-Sixth Annual SAS Users Group International Conference,* Paper 240-26. Cary, NC: SAS Institute Inc.

Hocking, R.R. (1985), *The Analysis of Linear Models*, Monterey, California: Brooks/Cole Publishing Company.

Lenth, R.V. (2001), "Some Practical Guidelines for Effective Sample Size Determination," *The American Statistician*, 55, 187–193.

Muller, K.E. and Benignus, V.A. (1992), "Increasing Scientific Power with Statistical Power," *Neurotoxicology and Teratology*, 14, 211–219.

Muller, K.E., LaVange, L.M., Ramey, S.L. and Ramey, C.T. (1992), "Power Calculations for General Linear Multivariate Models Including Repeated Measures Applications," *Journal of the American Statistical Association*, 87 (420), 1209–1226.

Muller, K.E. and Peterson, B.L. (1984), "Practical Methods for Computing Power in Testing the Multivariate General Linear Hypothesis," *Computational Statistics & Data Analysis*, 2, 143–158.

O'Brien, R.G. and Muller, K.E. (1993), "Unified Power Analysis for *t*-Tests Through Multivariate Hypotheses," in *Applied Analysis of Variance in Behavioral Science*, ed. L.K. Edwards, New York: Marcel Dekker, Chapter 8, 297–344.

O'Brien, R.G. and Shieh, G. (1992), "Pragmatic, Unifying Algorithm Gives Power Probabilities for Common F Tests of the Multivariate General Linear Hypothesis." Poster presented at the American Statistical Association Meetings, Boston, Statistical Computing Section. Also, paper in review, downloadable in PDF form from www.bio.ri.ccf.org/UnifyPow.